# Abrupt Changes and the Unit Step (Heaviside) Function

```
> restart;with(inttrans);
```

[*addtable*, *fourier*, *fouriercos*, *fouriersin*, *hankel*, *hilbert*, *invfourier*, *invhilbert*, *invlaplace*, *invmellin*, *laplace*, *mellin*, *savetable*]

We will begin by again plotting Heaviside $(t - 4)$.

```
> plot(Heaviside(t-4),t=-10..10, discont=true, thickness=3);
```




From the graph, it seems clear that $u(t-b)$ = Heaviside$(t-b) = \{0$ for $t < b$, 1 for $t > b\}$ and is undefined at $t = b$. This function helps us to deal with functions that are piecewise continuous, i.e., have finite jumps discontinuities at finitely many places. For instance, consider the piecewise continuous function

```
> f:=piecewise(t<0,0, t>0 and t<3,t,3<t and t<6,t+3,t>6,ln(t));
```

$$f := \begin{cases} 0 & t < 0 \\ t & 0 < t \text{ and } t < 3 \\ t+3 & 3 < t \text{ and } t < 6 \\ \ln(t) & 6 < t \end{cases}$$

We plot this function.

```
> plot(f,t=0..9,discont=true);
```




We rewrite the function using the unit step function and again look at its graph.

```
> g:=t*Heaviside(t)+3*Heaviside(t-3)+(ln(t)-t-3)*Heaviside(t-6);
```

$$g := t\,\text{Heaviside}(t) + 3\,\text{Heaviside}(-3+t) + (\ln(t) - t - 3)\,\text{Heaviside}(-6+t)$$

```
> plot(g,t=0..9,discont=true);
```




Now we look at the difference of the two functions.

```
> simplify(f-g);
```

$$\begin{cases} \mathit{undefined} & t=0 \\ \mathit{undefined} & t=3 \\ \mathit{undefined} & t=6 \\ 0 & \mathit{otherwise} \end{cases}$$

We see they are equal where defined.

Next we use the Laplace transform to solve the IVP $\frac{d}{dt}x(t)+\frac{3\,x(t)}{125}=48+24\,u(t-10)-24\,u(t-20)$, x(0)=1000. We illustrate with a step-by-step process.

```
> DE:=diff(x(t),t)+(3/125)*x(t)=48+24*Heaviside(t-10)-24*Heaviside(t-20);
```

$$DE := \frac{d}{dt}x(t)+\frac{3}{125}x(t)=48+24\,\mathrm{Heaviside}(t-10)-24\,\mathrm{Heaviside}(t-20)$$

```
> IC:=x(0)=1000;
```

$$IC := x(0)=1000$$

We take the Laplace transform of each side of the equation.

```
> L:=laplace(lhs(DE),t,s)=laplace(rhs(DE),t,s);
```

$$L := s\, laplace(x(t), t, s) - x(0) + \frac{3}{125}\, laplace(x(t), t, s) = \frac{24\left(2 + e^{-10\, s} - e^{-20\, s}\right)}{s}$$

We substitute in the value of the initial condition.

```
> LI:=subs(x(0)=1000,L);
```

$$LI := s\, laplace(x(t), t, s) - 1000 + \frac{3}{125}\, laplace(x(t), t, s) = \frac{24\left(2 + e^{-10\, s} - e^{-20\, s}\right)}{s}$$

We solve for $\boldsymbol{L}\{x\}$, denoting it by the variable **LT**, and then expanding the solution.

```
> LT:=solve(LI,laplace(x(t),t,s));
```

$$LT := \frac{1000\left(6 + 3\, e^{-10\, s} - 3\, e^{-20\, s} + 125\, s\right)}{s\,(125\, s + 3)}$$

```
> LT:=expand(LT);
```

$$LT := \frac{6000}{s\,(125\, s + 3)} + \frac{3000}{s\,(125\, s + 3)\left(e^{s}\right)^{10}} - \frac{3000}{s\,(125\, s + 3)\left(e^{s}\right)^{20}} + \frac{125000}{125\, s + 3}$$

Finally, we take the inverse Laplace transform to find our solution.

```
> soln_x(t):=invlaplace(LT,s,t);
```

$$soln\_x(t) := 2000 - 1000\, e^{-\frac{3}{125}\, t} + 1000\, \mathrm{Heaviside}(t - 10)\left(1 - e^{-\frac{3}{125}\, t + \frac{6}{25}}\right) - 1000\, \mathrm{Heaviside}(t - 20)\left(1 - e^{-\frac{3}{125}\, t + \frac{12}{25}}\right)$$

Of course, we could have used the simple Maple way.

```
> infolevel[dsolve]:=5;
```

$$infolevel_{dsolve} := 5$$

```
> dsolve({DE,IC},x(t));
Methods for first order ODEs:
--- Trying classification methods ---
trying a quadrature
trying 1st order linear
<- 1st order linear successful
```

$$x(t) = 2000 + 1000\, \mathrm{Heaviside}(t - 10) - 1000\, \mathrm{Heaviside}(t - 10)\, e^{-\frac{3}{125}\, t + \frac{6}{25}} - 1000\, \mathrm{Heaviside}(t - 20) + 1000\, \mathrm{Heaviside}(t - 20)\, e^{-\frac{3}{125}\, t + \frac{12}{25}} - 1000\, e^{-\frac{3}{125}\, t}$$

```
> dsolve({DE,IC},x(t),method=laplace);
dsolve/inttrans/solveit: Transform of eqns is {-1000-24*(2+exp(-10*_s1)
-exp(-20*_s1))/_s1+_s1*laplace/internal(x(t) t _s1)+(3/125)*
laplace/internal(x(t) t _s1)}
dsolve/inttrans/solveit: Algebraic Solution is {laplace/internal(x(t) t
_s1) = 1000*(125*_s1+6+3*exp(-10*_s1)-3*exp(-20*_s1))/(_s1*(125*_s1+3))
}
```

$$x(t) = 2000 - 1000\, e^{-\frac{3}{125}\, t} + 1000\, \mathrm{Heaviside}(t - 10)\left(1 - e^{-\frac{3}{125}\, t + \frac{6}{25}}\right) - 1000\, \mathrm{Heaviside}(t - 20)\left(1 - e^{-\frac{3}{125}\, t + \frac{12}{25}}\right)$$

We plot the solution.

```
> plot(soln_x(t),t=0..150);
```




It appears that x(t) approaches 2000 as a limit. We check that.

```
> limit(soln_x(t),t=infinity);
```

$$2000$$